# Advancing Lined Tunnel (Rigidly Connected to Zones)

Problem Statement

Note

To view this project in FLAC3D, use the menu command Help ‣ Examples…. Choose “Structure/Shell/AdvancingTunnel” and select “AdvancingTunnel.prj” to load. The main data files used are shown at the end of this example. The remaining data files can be found in the project.

This example demonstrates how to simulate the sequential operations of excavating and adding support to an advancing tunnel. The tunnel has a circular cross-section and is located at 5 m depth in a soft elastic soil ($$E$$ = 48.2 MPa, $$\nu$$ = 0.34) with isotropic in-situ stresses of 1 MPa. The tunnel is supported by shotcrete ($$E$$ = 10.5 GPa, $$\nu$$ = 0.25) with a thickness of 0.2 m. The shotcrete remains elastic and rigidly connected to the soil throughout the simulation. (The shotcrete/soil interface can be allowed to fail in either tension or shear, such that gaps can form and slip can occur, if liner elements are used instead of shells — see Advancing Lined Tunnel.)

The geometry for this model was constructed interactively using the Extruder tool, and the data file exported from the State Record into “geometry.dat”. Figure 1 shows the resulting geometry and the group names assigned to zones and zone faces using the Extruder tool and the zone face skin command.

Figure 1: Model geometry with zone and face names assigned.

Begin with an initial tunnel of 2 m total length. For simplicity, it is assumed that the excavation proceeds simultaneously on both tunnel faces; therefore, it is only necessary to model one quarter-section of the tunnel by applying symmetry boundary conditions on the three symmetry planes. The excavation process is modeled by assigning the null material model to zones. Then the stresses are allowed to redistribute. The shotcrete is now installed by creating shell elements and attaching them to the tunnel surface with the structure shell create by-zone-face command. The next tunnel segment is excavated and, again, the stresses are allowed to redistribute. The model at this stage is shown in Figure 2. This excavation sequence can be repeated to follow the entire tunnel construction.

Figure 2: Stage 2: Install shotcrete in section 1; excavate tunnel section 2.

Figure 3 and Figure 4 show the displacements that occur during the second excavation stage for the case with no support and the case with support, respectively. The $$z$$-displacement history at the tunnel crown is included in each figure. It can be seen that the support reduces the crown displacement from approximately 12 mm to 1 mm.

Figure 3: Displacements during stage 2—no support.

Figure 4: Displacements during stage 2—shotcrete support.

The bending stress resultant, $$M_x$$, where the $$x$$-direction lies along the tunnel axis, is shown in Figure 5. It can be seen that bending is most extreme at the front of the liner near the tunnel face. A plot of the deformed shape of the liner in Figure 6 indicates that this intense bending arises from the large pinching deformation.

Figure 5: Bending stress resultant, $$M_x$$, at end of stage 2.

Figure 6: Deformed (magnification of 150) shape of shotcrete at end of stage 2.

One additional excavation stage is performed. The model at the end of this stage is shown in Figure 7. The shotcrete support has been extended into tunnel section 2 by issuing another structure shell create by-zone-face command, noting that by default a new ID number is issued for the elements created. This creates a “cold-joint” between the two shotcrete segments. (There will be a collection of double nodes lying along the interface: one is used by segment one; the other is used by segment two. Forces and moments will not be transferred between the adjoining elements; instead, only forces will be transmitted into the surrounding zones at the common locations. This mimics two separate shell segments lying next to one another.) The deformation that occurs during stage 3 begins to load the new shell segment, and produces additional load in the previous shell segment. Both the displacement (see Figure 8) and the bending stress (see Figure 9) fields are discontinuous across the joint.

Figure 7: Stage 3: Install shotcrete in section 2; excavate tunnel section 3.

Figure 8: Deformed (magnification of 150) shape of shotcrete at end of stage 3.

Figure 9: Bending stress resultant, $$M_x$$, at end of stage 3.

Data File

model new
model large-strain off
model title 'Advancing Lined Tunnel (Rigidly Connected to Zones)'
; Create simple zone geometry, created interactively in Extruder
; and exported from the State Record
program call 'geometry' suppress
; Also assigns group names to model zones and faces.
zone generate from-extruder
zone face skin ; Label model boundaries
; Model and properties
zone cmodel assign elastic
zone property bulk 50e6 shear 18e6
; Initial conditions
zone initialize stress xx -1e6 yy -1e6 zz -1e6
; Boundary Conditions - roller on near boundaries and fixed on far.
zone face apply velocity-normal 0 range group 'West'
zone face apply velocity-normal 0 range group 'South'
zone face apply velocity-normal 0 range group 'Bottom'
zone face apply velocity (0,0,0) range group 'East' or 'North' or 'Top'
; --- Stage 1: excavate tunnel section 1
zone cmodel assign null range group 'section1' group 'tunnel'
model solve convergence 1
model save 'Stage1'
; Reset displacements, take history of roof closure
zone gridpoint initialize displacement (0,0,0)
zone history displacement-z position (0,1,1)
; --- Stage 2: excavate tunnel section 2
zone cmodel assign null range group 'section2' group 'tunnel'
model save 'Stage2-start'
model solve convergence 1
model save 'Stage2-nosupport'

; -- Try with support
model restore 'Stage2-start'
; --- Add shotcrete, on shotcrete faces in section 1.
struct shell create by-zone-face group 'shotcrete1' ...
range group 'shotcrete' group 'section1'
struct shell property isotropic=(10.5e9, 0.25) thickness=0.2
; --- Boundary conditions for structural elements
struct node fix velocity-x rotation-y rotation-z range position-x 0
; symmetry conditions
struct node fix velocity-y rotation-x rotation-z range position-y 0
struct node fix velocity-z rotation-x rotation-y range position-z 0
model solve convergence 1
model save 'Stage2-support'
; --- Stage 3: excavate tunnel section 3
zone cmodel assign null range group 'section3' group 'tunnel'
; --- Add shotcrete, on shotcrete faces in section 1.
struct shell create by-zone-face group 'shotcrete2' ...
range group 'shotcrete' group 'section2'
struct shell property isotropic=(10.5e9, 0.25) thickness=0.2
; --- Boundary conditions for structural elements
struct node fix velocity-x rotation-y rotation-z range position-x 0
; symmetry conditions
struct node fix velocity-y rotation-x rotation-z range position-y 0
struct node fix velocity-z rotation-x rotation-y range position-z 0
model solve convergence 1
model save 'Stage3-support'